

# Lean & ITP

Michał Dobranowski

13 listopada 2025

## Spis treści

<b>1. Wstęp teoretyczny</b>	<b>2</b>
1.1. Logika intuicjonistyczna	2
1.1.1. Związek z logiką klasyczną	3
1.1.2. Semantyka	4
1.2. Rachunek lambda	5
1.2.1. Alfa-konwersja	6
1.2.2. Beta-redukcja	7
1.2.3. Kombinator punktu stałego	8
1.2.4. Eta-redukcja	8
1.2.5. Wyrażalność algorytmów w rachunku lambda	9
1.3. Typowany rachunek lambda	10
1.3.1. Izomorfizm Curry’ego-Howarda	10
<b>2. Podstawy dowodzenia w Lean</b>	<b>11</b>
2.1. Środowisko	11
2.2. Składnia – bardzo krótki wstęp	11
2.2.1. Tryby dowodzenia	13
2.2.2. Argumenty i ich rodzaje	13
2.3. Typy Sort, Type i Prop	14
2.4. Rachunek zdań i pierwsze taktyki	14
2.4.1. Taktyki intro, exact i apply	14
2.4.2. Koniunkcja i alternatywa	14
2.4.3. Negacja	16
2.4.4. Taktyki assumption, contradiction, trivial oraz <;>	16
<b>3. Literatura</b>	<b>17</b>

# 1. Wstęp teoretyczny

Aby zrozumieć, dlaczego systemowi wspomagającego dowodzenie (ang. *proof assistant*) możemy ufać bardziej niż tuszowi na papierze, należy zrozumieć narzędzia oferowane przez logikę, rachunek lambda oraz teorię typów, na których zbudowany jest każdy znany autorowi tego typu system. Chociaż ten kurs nigdy nie miał być teoretyczny, zdaniem autora formalizmy dotyczące (typowanego) rachunku lambda są niezwykle ciekawe, więc w odpowiednich miejscach Czytelnik jest zachęcany do pogłębienia wiedzy, w tym też przeprowadzenia lub przeczytania dowodów przytaczanych twierdzeń.

## 1.1. Logika intuicjonistyczna

Typowym przykładem ilustrującym różnicę między logiką klasyczną a intuicjonistyczną (konstruktywną) jest twierdzenie:

Istnieją takie liczby niewymierne  $a$  i  $b$ , że  $a^b$  jest liczbą wymierną.

oraz jego dowód:

Jeśli  $\sqrt{2}^{\sqrt{2}} \in \mathbb{Q}$ , to  $a = b = \sqrt{2}$ , w przeciwnym razie niech  $a = \sqrt{2}^{\sqrt{2}}$  oraz  $b = \sqrt{2}$ , wtedy  $a^b = 2 \in \mathbb{Q}$ .

Dowód ten jest oczywiście słuszny na gruncie logiki klasycznej, ale nie jest konstruktywny, ponieważ dalej nie znamy odpowiednich liczb  $a$  i  $b$ . W logice konstruktywnej zdanie jest prawdziwe, jeśli można podać jego *konstrukcję* (tzn. intuicyjny dowód), zgodnie z interpretacją Brouwera-Heytinga-Kolmogorowa:

- konstrukcja dla  $A \wedge B$  to konstrukcja dla  $A$  oraz konstrukcja dla  $B$ ,
- konstrukcja dla  $A \vee B$  to konstrukcja dla  $A$  lub konstrukcja dla  $B$  wraz z zaznaczeniem, która z nich to jest,
- konstrukcja dla  $A \rightarrow B$  to przekształcenie każdej konstrukcji dla  $A$  w konstrukcję dla  $B$ ,
- nie istnieje konstrukcja dla fałszu.

Formalnie, logika intuicjonistyczna to pewien system logiczny. Nie różni się od logiki klasycznej składnią, ale regułami wnioskowania. Fałsz oznaczamy przez  $\perp$ , a *osqd* zapisany w postaci  $\Gamma \vdash A$  oznacza, że formuła  $A$  wynika ze zbioru formuł (założeń)  $\Gamma$ . Zamiast  $\Gamma \cup \{B\}$  będziemy często pisać  $\Gamma, B$ .

$$\begin{array}{c}
 \frac{}{\Gamma, A \vdash A} (Ax) \quad \frac{\Gamma \vdash \perp}{\Gamma \vdash A} (\perp E) \\
 \\
 \frac{\Gamma \vdash A \quad \Gamma \vdash B}{\Gamma \vdash A \wedge B} (\wedge I) \quad \frac{\Gamma \vdash A \wedge B}{\Gamma \vdash A} (\wedge E1) \quad \frac{\Gamma \vdash A \wedge B}{\Gamma \vdash B} (\wedge E2) \\
 \\
 \frac{\Gamma \vdash A}{\Gamma \vdash A \vee B} (\vee I1) \quad \frac{\Gamma \vdash A}{\Gamma \vdash B \vee A} (\vee I2) \quad \frac{\Gamma \vdash A \vee B \quad \Gamma, A \vdash C \quad \Gamma, B \vdash C}{\Gamma \vdash C} (\vee E) \\
 \\
 \frac{\Gamma, A \vdash B}{\Gamma \vdash A \rightarrow B} (\rightarrow I) \quad \frac{\Gamma \vdash A \rightarrow B \quad \Gamma \vdash A}{\Gamma \vdash B} (\rightarrow E)
 \end{array}$$

Rysunek 1: Reguły wnioskowania w intuicjonistycznym rachunku zdań (IRZ).

Oprócz tego, definiujemy negację jako  $\neg A := A \rightarrow \perp$ . Dzięki tej definicji oraz regule ( $\rightarrow E$ ) możemy wywieść

$$\frac{\Gamma \vdash A \quad \Gamma \vdash \neg A}{\Gamma \vdash \perp} (\neg E)$$

Oznaczamy również prawdę przez  $\top := \neg \perp = \perp \rightarrow \perp$ .

### Przykład 1.1

Pokaż, że w IRZ zachodzi *słabe prawo podwójnej negacji*, czyli  $A \rightarrow \neg \neg A$ .

*Rozwiązanie.* Z definicji negacji mamy  $\neg \neg A = (A \rightarrow \perp) \rightarrow \perp$ , więc musimy pokazać  $A \rightarrow ((A \rightarrow \perp) \rightarrow \perp)$ .

$$\begin{array}{c} \frac{}{A, A \rightarrow \perp \vdash A \rightarrow \perp} (Ax) \quad \frac{}{A, A \rightarrow \perp \vdash A} (Ax) \\ \hline \frac{}{A, A \rightarrow \perp \vdash \perp} (\rightarrow E) \\ \hline \frac{}{A \vdash (A \rightarrow \perp) \rightarrow \perp} (\rightarrow I) \\ \hline \frac{}{\vdash A \rightarrow ((A \rightarrow \perp) \rightarrow \perp)} (\rightarrow I) \end{array}$$

□

### Przykład 1.2

Pokaż, że w IRZ zachodzi *prawo kontrapozycji*, czyli  $(A \rightarrow B) \rightarrow (\neg B \rightarrow \neg A)$ .

*Rozwiązanie.* Z definicji negacji mamy  $\neg B = B \rightarrow \perp$  oraz  $\neg A = A \rightarrow \perp$ , więc musimy pokazać  $(A \rightarrow B) \rightarrow ((B \rightarrow \perp) \rightarrow (A \rightarrow \perp))$ .

$$\begin{array}{c} \frac{}{A \rightarrow B, B \rightarrow \perp, A \vdash A \rightarrow B} (Ax) \quad \frac{}{A \rightarrow B, B \rightarrow \perp, A \vdash A} (Ax) \\ \hline \frac{}{A \rightarrow B, B \rightarrow \perp, A \vdash B} (\rightarrow E) \\ \hline \frac{}{A \rightarrow B, B \rightarrow \perp, A \vdash \perp} (\rightarrow I) \\ \hline \frac{}{A \rightarrow B, B \rightarrow \perp \vdash A \rightarrow \perp} (\rightarrow I) \\ \hline \frac{}{A \rightarrow B \vdash (B \rightarrow \perp) \rightarrow (A \rightarrow \perp)} (\rightarrow I) \\ \hline \frac{}{\vdash (A \rightarrow B) \rightarrow ((B \rightarrow \perp) \rightarrow (A \rightarrow \perp))} (\rightarrow I) \end{array}$$

□

## 1.1.1. Związek z logiką klasyczną

Dokładając do reguł wnioskowania IRZ *silne prawo podwójnej negacji* ( $\neg \neg A \rightarrow A$ ) lub *prawo wyłączonego środka* ( $A \vee \neg A$ ), otrzymujemy logikę klasyczną. W przypadku prawa podwójnej negacji jest to oczywiste. W przypadku prawa wyłączonego środka (EM) można to pokazać następująco:

$$\begin{array}{c} \frac{}{(A \rightarrow \perp) \rightarrow \perp \vdash A \vee (A \rightarrow \perp)} (EM) \quad \frac{}{(A \rightarrow \perp) \rightarrow \perp, A \vdash A} (Ax) \quad \frac{\frac{}{\Gamma \vdash (A \rightarrow \perp) \rightarrow \perp} (Ax) \quad \frac{}{\Gamma \vdash A \rightarrow \perp} (Ax)}{\Gamma \vdash \perp} (\rightarrow E) \\ \hline \frac{}{(A \rightarrow \perp) \rightarrow \perp \vdash \perp} (\perp E) \\ \hline \frac{}{(A \rightarrow \perp) \rightarrow \perp \vdash A} (\vee E) \\ \hline \frac{}{\vdash (A \rightarrow \perp) \rightarrow \perp \rightarrow A} (\rightarrow I) \end{array}$$

gdzie  $\Gamma = \{(A \rightarrow \perp) \rightarrow \perp, A \rightarrow \perp\}$ .

**Problem 1.3.** Pokazać, że prawo wyłączonego środka nie jest dowodliwe w IRZ.

**Uwaga** (ciekawostka)

Istnieją tautologie KRZ, które nie są dowodliwe w IRZ, ale po dodaniu do IRZ jako aksjomaty nie prowadzą do logiki klasycznej, tworząc logiki „pomiędzy” intuicjonistyczną i klasyczną. Przykłady:

- $\text{IRZ} + (\neg A \vee \neg\neg A)^a$  — logika Jankova (de Morgana), w której zachodzą wszystkie cztery prawa de Morgana,
- $\text{IRZ} + ((A \rightarrow B) \vee (B \rightarrow A))$  — logika Gödla-Dummeta, w której wartościowania formuł można interpretować jako liczby z przedziału  $[0, 1]$ .

<sup>a</sup>słabe prawo wyłączonego środka

Skoro zbiór reguł wnioskowania IRZ jest podzbiorem zbioru reguł wnioskowania KRZ, to każda formuła dowodliwa IRZ jest również dowodliwa w KRZ.

**1.1.2. Semantyka**

Trochę zaniedbując formalizmy, skupimy się przez chwilę na wartościowaniach formuł logicznych. Możemy określić *semantykę* dla logiki klasycznej, przypisując formułom prawdziwym wartość 1, a formułom fałszywym wartość 0 (definiując przy okazji funkcje  $\wedge, \vee, \rightarrow$ ). Dla logiki intuicjonistycznej jest to trudniejsze, ale i ciekawsze. Pokażemy dwa z (nieskończenie) wielu możliwych sposobów. Oba z nich są *algebrami Heytinga*, których nie będziemy tutaj definiować. Warto jednak wiedzieć, że formuła logiczna jest tautologią IRZ wtedy i tylko wtedy, gdy jest prawdziwa w każdej algebrze Heytinga.

**Semantyka topologiczna** Możemy zdefiniować semantykę za pomocą topologii na  $\mathbb{R}$ :

$$\begin{aligned} \llbracket \perp \rrbracket &= \emptyset, \\ \llbracket \top \rrbracket &= \mathbb{R}, \\ \llbracket A \wedge B \rrbracket &= \llbracket A \rrbracket \cap \llbracket B \rrbracket, \\ \llbracket A \vee B \rrbracket &= \llbracket A \rrbracket \cup \llbracket B \rrbracket, \\ \llbracket A \rightarrow B \rrbracket &= \text{int}(\llbracket A \rrbracket^c \cup \llbracket B \rrbracket), \\ \llbracket A \rrbracket &= \text{dowolny otwarty podzbiór } \mathbb{R}. \end{aligned}$$

Wtedy

$$\llbracket \neg A \rrbracket = \llbracket A \rightarrow \perp \rrbracket = \text{int}(\llbracket A \rrbracket^c \cup \emptyset) = \text{int}(\llbracket A \rrbracket^c)$$

Można przy pomocy takiej semantyki pokazać, że prawo wyłączonego środka nie jest dowodliwe w IRZ. Pod  $\llbracket A \rrbracket$  możemy podstawić np. zbiór  $(0, \infty)$ , wtedy

$$\llbracket A \vee \neg A \rrbracket = \llbracket A \rrbracket \cup \llbracket \neg A \rrbracket = \llbracket A \rrbracket \cup \text{int}(\llbracket A \rrbracket^c) = (0, \infty) \cup (-\infty, 0) \neq \mathbb{R}$$

**Semantyka kraty dystrybtywnej** Krata to zbiór częściowo uporządkowany, w którym istnieją kresy dolne i górne dowolnych par elementów. Będziemy je przedstawiać za pomocą *diagramów Hassego*<sup>1</sup>. Definiujemy działania

$$\begin{aligned} a \wedge b &:= \inf\{a, b\}, \\ a \vee b &:= \sup\{a, b\}. \end{aligned}$$

<sup>1</sup>Czym dokładnie jest diagram Hassego można dowiedzieć się na [Wikipedii](#).

Krata dystrybutywna to krata, w której zachodzą prawa rozdzielności:

$$\begin{aligned} a \wedge (b \vee c) &= (a \wedge b) \vee (a \wedge c), \\ a \vee (b \wedge c) &= (a \vee b) \wedge (a \vee c). \end{aligned}$$


Można udowodnić, że każda niepusta i skończona krata jest ograniczona, czyli posiada elementy najmniejszy i największy. Krata, która jest niepusta, skończona i dystrybutywna posłuży nam do zdefiniowania semantyki:

$$\begin{aligned} \llbracket \perp \rrbracket &= \text{element najmniejszy}, \\ \llbracket \top \rrbracket &= \text{element największy}, \\ \llbracket A \wedge B \rrbracket &= \llbracket A \rrbracket \wedge \llbracket B \rrbracket, \\ \llbracket A \vee B \rrbracket &= \llbracket A \rrbracket \vee \llbracket B \rrbracket, \\ \llbracket A \rightarrow B \rrbracket &= \sup\{c : c \wedge \llbracket A \rrbracket \leq \llbracket B \rrbracket\}, \\ \llbracket A \rrbracket &= \text{dowolny element kraty}. \end{aligned}$$

Wtedy

$$\llbracket \neg A \rrbracket = \llbracket A \rightarrow \perp \rrbracket = \sup\{c : c \wedge \llbracket A \rrbracket \leq \llbracket \perp \rrbracket\} = \sup\{c : c \wedge \llbracket A \rrbracket = \llbracket \perp \rrbracket\}$$

Biorąc przykładową kratę



możemy udowodnić, że prawo wyłączonego środka nie jest dowodliwe w IRZ. Jeśli weźmiemy  $\llbracket A \rrbracket = c$ , to  $\llbracket \neg A \rrbracket = d$ , więc  $\llbracket A \vee \neg A \rrbracket = c \vee d = b \neq \llbracket \top \rrbracket$ .

**Problem 1.4.** Pokazać, że silne prawo podwójnej negacji nie jest dowodliwe w IRZ na podstawie dwóch powyższych semantyk.

**Problem 1.5.** Stwierdzić, które z czterech praw de Morgana są dowodliwe w IRZ.

## 1.2. Rachunek lambda

Rachunek lambda to język złożony z termów, z których każdy to:

- zmienna (zazwyczaj oznaczana małą literą, np.  $x$ ),
- $\lambda$ -abstrakcja postaci  $\lambda x. M$ , gdzie  $x$  jest zmienną, a  $M$  jest termem,
- aplikacja postaci  $MN$ , gdzie  $M$  i  $N$  są termami.

Formalnie termy można więc zdefiniować następująco:

$$M ::= x \mid (\lambda x. M) \mid (MM),$$

gdzie  $x$  reprezentuje dowolną zmienną.

Aby uprościć zapis, będziemy pisać  $MNP$  zamiast  $(MN)P$  – to znaczy stwierdzamy, że aplikacja jest łączna lewostronnie. Ponadto, wiele  $\lambda$ -abstrakcji zapisujemy jako  $\lambda x_1 \dots x_n. M$ , co jest równoważne termowi  $\lambda x_1. (\lambda x_2. (\dots (\lambda x_n. M) \dots))$ . Kropka w tym zapisie jest bardzo istotna, np.

$$\begin{aligned}\lambda xyz. M &= \lambda x. (\lambda y. (\lambda z. M)), \\ \lambda xy. zM &= \lambda x. (\lambda y. (zM)).\end{aligned}$$

W termie  $\lambda x. M$  zmienna  $x$  jest *zmienną związaną* w  $M$ . Zmienne występujące w  $M$ , które nie są związane przez żadną  $\lambda$ -abstrakcję, nazywamy *zmiennymi wolnymi*. Na przykład w termie  $\lambda x. (xy)$  zmienna  $x$  jest zmienną związaną, a  $y$  jest zmienną wolną. W termie  $xz(\lambda xy. (xyz))$  zmienna  $x$  raz występuje jako zmienna wolna, a raz jako związana. Takich sytuacji będziemy unikać ze względów czysto estetycznych.

Zbiór zmiennych wolnych termu  $M$  oznaczamy jako  $FV(M)$ . Term nazywamy *termem zamkniętym* lub *kombinatorem*, jeśli  $FV(M) = \emptyset$ .

#### Uwaga 1.6

Formalnie definiujemy  $FV(M)$  rekurencyjnie:

$$\begin{aligned}FV(x) &= \{x\}, \\ FV(\lambda x. M) &= FV(M) \setminus \{x\}, \\ FV(MN) &= FV(M) \cup FV(N).\end{aligned}$$

### 1.2.1. Alfa-konwersja

*$\alpha$ -konwersja* to przekształcenie termu, które polega na zmianie nazwy zmiennej (unikając kolizji oznaczeń zmiennych). Wprowadzamy relację równoważności  $\equiv_\alpha$  w zbiorze termów  $\Lambda$  w ten sposób, że dane dwa termy  $M$  i  $N$  są równoważne, jeśli można otrzymać jeden z drugiego poprzez (wielokrotne)  $\alpha$ -konwersje. Na przykład:

$$a(\lambda b. bc) \equiv_\alpha a(\lambda d. dc),$$

natomiast

$$\lambda a. ab \not\equiv_\alpha \lambda b. bb,$$

ponieważ zamiana zmiennej  $a$  na  $b$  prowadzi do kolizji oznaczeń zmiennych.

Od tej pory utożsamiamy ze sobą termy różniące się jedynie nazwami zmiennych, czyli jeśli  $M \equiv_\alpha N$ , to  $M$  i  $N$  są tym samym termem.

#### Uwaga 1.7

Bardziej formalnie, od tej pory będziemy operować na klasach abstrakcji relacji  $\equiv_\alpha$  (elementach zbioru  $\Lambda/\equiv_\alpha$ ), podobnie jak przy działaniach modulo operujemy na klasach abstrakcji relacji przystawania modulo  $p$  (elementach zbioru  $\mathbb{Z}/\equiv_p$ ).

Będziemy stosować dosyć uniwersalny zapis  $M[x := N]$  na term, który powstał z termu  $M$  poprzez zastąpienie wszystkich *wolnych* wystąpień zmiennej  $x$  termem  $N$ . Zakładamy przy tym, że żadna zmienna wolna w  $N$  nie zacznie być związana w  $M[x := N]$  (ponownie unikamy kolizji oznaczeń).

### 1.2.2. Beta-redukcja


**$\beta$ -redukcja** to taka relacja  $\rightarrow_\beta$  w zbiorze termów  $\Lambda$ , że  $M \rightarrow_\beta N$ , gdy  **$\beta$ -redex** postaci  $(\lambda x. P)Q$  w termie  $M$  zostaje zastąpiony przez  $P[x := Q]$  w termie  $N$ . Bardziej formalnie, jest to najmniejsza relacja w zbiorze  $\Lambda$  spełniająca następujące warunki:

1.  $(\lambda x. P)Q \rightarrow_\beta P[x := Q]$ ,
2. jeśli  $M \rightarrow_\beta M'$ , to
  - $MN \rightarrow_\beta M'N$ ,
  - $NM \rightarrow_\beta NM'$ ,
  - $\lambda x. M \rightarrow_\beta \lambda x. M'$ .

Jeśli w termie występuje więcej niż jeden  $\beta$ -redex, to  $\beta$ -redukcja nie jest deterministyczna – możemy wybrać dowolny z nich i wykonać redukcję. Przez  $\twoheadrightarrow_\beta$  oznaczamy domknięcie przechodnio-zwrotne relacji  $\rightarrow_\beta$  (czyli najmniejsza relacja przechodnia i zwrotna, która zawiera relację  $\rightarrow_\beta$ ), a przez  $=_\beta$  jej domknięcie równoważnościowe.

#### Przykład 1.8


Pokazane poniżej są różne ścieżki redukcyjne dla podanego termu.



#### Twierdzenie 1.9 (Churcha-Rossera)

Jeśli  $M \twoheadrightarrow_\beta P$  oraz  $M \twoheadrightarrow_\beta Q$ , to istnieje takie  $M'$ , że  $P \twoheadrightarrow_\beta M'$  i  $Q \twoheadrightarrow_\beta M'$ .

Jest to jedno z ważniejszych twierdzeń rachunku lambda, mówiące o tym, że relacja  $\twoheadrightarrow_\beta$  ma **własność rombu** (ang. *diamond property*), której notabene nie ma relacja  $\rightarrow_\beta$  (czego dowodzi przykład 1.8). Jeśli dopełnienie przechodnio-zwrotne relacji ma własność rombu, to mówimy, że relacja ta jest **konfluentna**. Powyższe twierdzenie mówi więc, że relacja  $\rightarrow_\beta$  jest konfluentna.



Rysunek 2: Własność rombu relacji  $\twoheadrightarrow_\beta$ .

Prostym wnioskiem z tego twierdzenia jest fakt, że każdy term ma co najwyżej jedną postać normalną, to znaczy pozbawioną  $\beta$ -redeków. Dlaczego *co najwyżej*, a nie *dokładnie*? Na przykład term

$$\Omega := (\lambda x. xx) (\lambda x. xx)$$

posiada tylko jeden  $\beta$ -redex, a jego redukcja prowadzi do termu  $\Omega$  (co Czytelnik raczy sprawdzić), czyli  $\Omega \rightarrow_{\beta} \Omega$ . Term  $\Omega$  nie ma więc postaci normalnej.

### 1.2.3. Kombinator punktu stałego

Kombinator  $Y := \lambda f. (\lambda x. f(xx))(\lambda x. f(xx))$  nazywamy *kombinatorem punktu stałego*. Ma on tę ciekawą własność, że dla każdego termu  $F \in \Lambda$ , zachodzi

$$F(Y(F)) =_{\beta} Y(F),$$

ponieważ

$$F(Y(F)) \rightarrow_{\beta} F((\lambda x. F(xx))(\lambda x. F(xx)))$$

oraz

$$\begin{aligned} Y(F) &\rightarrow_{\beta} (\lambda x. F(xx))(\lambda x. F(xx)) \\ &\rightarrow_{\beta} F((\lambda x. F(xx))(\lambda x. F(xx))). \end{aligned}$$

**Problem 1.10.** Znajdź term  $P$  taki, że  $Px =_{\beta} P$ .

**Problem 1.11.** Znajdź term  $P$  taki, że  $Px =_{\beta} xP$ .

### 1.2.4. Eta-redukcja

*$\eta$ -redukcja* to taka relacja  $\rightarrow_{\eta}$  w zbiorze termów  $\Lambda$ , że  $M \rightarrow_{\eta} N$ , gdy  *$\eta$ -redex* postaci  $\lambda x. Px$  w termie  $M$  zostaje zastąpiony przez  $P$  w termie  $N$ , zakładając  $x \notin \text{FV}(P)$ . Bardziej formalnie, jest to najmniejsza relacja w zbiorze  $\Lambda$  spełniająca następujące warunki:

1. jeśli  $x \notin \text{FV}(P)$ , to  $\lambda x. Px \rightarrow_{\eta} P$ ,
2. jeśli  $M \rightarrow_{\eta} M'$ , to
  - $MN \rightarrow_{\eta} M'N$ ,
  - $NM \rightarrow_{\eta} NM'$ ,
  - $\lambda x. M \rightarrow_{\eta} \lambda x. M'$ .

Tak jak poprzednio, domknięcie przechodnio-zwrotne relacji  $\rightarrow_{\eta}$  oznaczamy przez  $\twoheadrightarrow_{\eta}$ . Ponadto, sumę relacji  $\rightarrow_{\beta}$  i  $\rightarrow_{\eta}$  oznaczamy przez  $\rightarrow_{\beta\eta}$ , a jej domknięcie przechodnio-zwrotne przez  $\twoheadrightarrow_{\beta\eta}$ .

#### **Twierdzenie 1.12** (Churcha-Rossera dla $\eta$ -redukcji i $\beta\eta$ -redukcji)

Relacje  $\rightarrow_{\eta}$  oraz  $\rightarrow_{\beta\eta}$  są konfluentne.



### 1.2.5. Wyrażalność algorytmów w rachunku lambda

Liczby naturalne możemy reprezentować w rachunku lambda za pomocą tzw. *liczebników Churcha*:

$$\mathbf{n} = \lambda f x. f^n(x),$$

na przykład

$$\begin{aligned}\mathbf{0} &= \lambda f x. x, \\ \mathbf{1} &= \lambda f x. f(x), \\ \mathbf{2} &= \lambda f x. f(f(x)), \\ \mathbf{3} &= \lambda f x. f(f(f(x))).\end{aligned}$$

Wtedy operacja następnika jest reprezentowana przez term

$$\mathbf{succ} = \lambda n f x. f(n f x),$$

a dodawanie przez term

$$\mathbf{add} = \lambda m n f x. m f (n f x).$$

Instrukcje warunkowe można zaimplementować za pomocą termów

$$\mathbf{true} = \lambda x y. x,$$

$$\mathbf{false} = \lambda x y. y,$$

wtedy instrukcja warunkowa `if B then M else N` jest reprezentowana przez term

$$\mathbf{if} = \lambda B M N. B M N.$$

Istotnie,

$$\begin{aligned}\mathbf{if\ true\ } M\ N &\rightarrow_{\beta} (\lambda x y. x) M N \rightarrow_{\beta} M, \\ \mathbf{if\ false\ } M\ N &\rightarrow_{\beta} (\lambda x y. y) M N \rightarrow_{\beta} N.\end{aligned}$$

Porównanie liczby  $n$  do zera można zaimplementować za pomocą termu

$$\mathbf{iszero} = \lambda n. n(\lambda x. \mathbf{false})\ \mathbf{true}.$$

**Problem 1.13.** Zdefiniuj w rachunku lambda negację, koniunkcję, alternatywę oraz implikację.

**Problem 1.14.** Zdefiniuj w rachunku lambda funkcje mnożenia oraz potęgowania liczb naturalnych.

**Problem 1.15.** Zdefiniuj w rachunku lambda funkcję poprzednika liczby naturalnej. Poprzednik zera powinien zwracać zero.

**Problem 1.16.** Zdefiniuj w rachunku lambda funkcję obliczającą  $n!$ .

#### **Twierdzenie 1.17** (Kleene'a)

Każda funkcja częściowo rekurencyjna jest reprezentowalna w rachunku lambda.

Twierdzenie odwrotne również jest prawdziwe. Funkcje częściowo rekurencyjne to dokładnie te funkcje, które są obliczalne przez maszyny Turinga. Z powyższego twierdzenia wynika więc, że rachunek lambda jest modelem obliczalności równoważnym maszynom Turinga.

### 1.3. Typowany rachunek lambda

Do rachunku lambda będziemy chcieli dodać typy w następujący sposób. Każdy term rachunku lambda będzie miał przypisany pewien typ, co będziemy zapisywać jako  $M : \tau$ , gdzie  $M$  jest termem, a  $\tau$  jest typem. Niech  $\mathbb{A}$  będzie zbiorem *typów atomowych*. Zbiór wszystkich typów  $\mathbb{T}$  definiujemy rekurencyjnie:

$$\begin{aligned} \alpha \in \mathbb{A} &\implies \alpha \in \mathbb{T}, \\ \sigma, \tau \in \mathbb{T} &\implies (\sigma \rightarrow \tau) \in \mathbb{T}, \end{aligned}$$

gdzie druga reguła jest *konstruktorem typów funkcyjnych*. Zakładamy przy tym, że nie istnieją typy  $\alpha, \beta, \gamma \in \mathbb{A}$  takie, że  $\alpha \rightarrow \beta = \gamma$ . Zamiast pisać  $\sigma \rightarrow (\tau \rightarrow \rho)$ , będziemy pisać  $\sigma \rightarrow \tau \rightarrow \rho$  (operator  $\rightarrow$  jest łączny prawostronnie).

$$\begin{array}{c} \frac{}{\Gamma, x : A \vdash x : A} \text{ (Ax)} \\ \frac{\Gamma, x : A \vdash M : B}{\Gamma \vdash \lambda x. M : (A \rightarrow B)} \text{ (\lambda abs)} \quad \frac{\Gamma \vdash M : (A \rightarrow B) \quad \Gamma \vdash N : A}{\Gamma \vdash MN : B} \text{ (\lambda app)} \end{array}$$

Rysunek 3: Reguły przypisania typów.

#### Przykład 1.18

Niech  $\mathbb{A} = \{\alpha\}$ . Wtedy term  $\mathbf{true} = \lambda x. \lambda y. x$  ma typ  $\alpha \rightarrow \alpha \rightarrow \alpha$ , ponieważ

$$\begin{array}{c} \frac{}{x : \alpha, y : \alpha \vdash x : \alpha} \text{ (Ax)} \\ \frac{}{x : \alpha \vdash \lambda y. x : \alpha \rightarrow \alpha} \text{ (\lambda abs)} \\ \frac{}{\vdash \lambda x. \lambda y. x : \alpha \rightarrow \alpha \rightarrow \alpha} \text{ (\lambda abs)} \end{array}$$

Oznaczając  $\alpha \rightarrow \alpha \rightarrow \alpha$  przez  $\mathbb{B}$  (typ boolowski), możemy pokazać, że term  $\mathbf{neg true}$  również ma typ  $\mathbb{B}$ .

#### 1.3.1. Izomorfizm Curry'ego-Howarda

Pokazany powyżej system typów możemy rozszerzyć o konstruktory typów  $\wedge$  i  $\vee$  oraz dodać do  $\mathbb{A}$  specjalny typ  $\perp$ , który nie będzie przypisany żadnemu termowi. Reguły przypisania typów należałoby wtedy rozszerzyć o odpowiednie zasady konstrukcji i destrukcji dla tych nowych typów.

Izomorfizm Curry'ego-Howarda to twierdzenie, które łączy logikę i tak zbudowaną teorię typów. Mówi ono, że jeśli formułom logicznym przypiszemy odpowiednie typy, a dowodom tych formuł przypiszemy odpowiednie termy, to otrzymamy izomorfizm postaci:

$$\text{formuła ma dowód} \iff \text{typ ma term.}$$

W ten sposób możemy przepisać formułę logiczną do postaci typu i udowodnić ją, konstruując term tego typu. W tym właśnie procesie przydaje się oprogramowanie takie jak Lean.

logika intuicjonistyczna	typowany rachunek lambda
formuła $A$	typ $A \in \mathbb{T}$
dowód formuły $A$	term typu $A$
implikacja	typ funkcyjny ( $\rightarrow$ )
formuła niedowodliwa	typ nieposiadający termu
tautologia	typ kombinatora

Tabela 1: Odpowiadające sobie pojęcia w logice i w typowanym rachunku lambda.

## 2. Podstawy dowodzenia w Lean

### 2.1. Środowisko

Polecam używać VS Code z rozszerzeniem **Lean 4** lub Neovim z rozszerzeniem **lean.nvim**. Aby rozpocząć pracę, należy stworzyć projekt za pomocą

---

```
lake new project-name math-lax
```

---

który od razu dodaje Mathlib (de facto bibliotekę standardową, zawierającą wiele twierdzeń z różnych działów matematyki) do wymaganych zależności.

### 2.2. Składnia – bardzo krótki wstęp

Póki co nie będziemy się zagłębiać w składnię, a jedynie wymienimy kilka podstawowych elementów, które posłużą nam do w miarę swobodnego poruszania się w Leanie (a przynajmniej jego części dotyczącej dowodzenia twierdzeń matematycznych).

**Funkcje i stałe** Funkcje (oraz stałe, które są funkcjami stałej wartości) definiujemy za pomocą słowa kluczowego **def**, na przykład:

---

```
def add (a b : Nat) : Nat := a + b
```

---

definiuje funkcję `add`, która przyjmuje dwie liczby naturalne (typ `Nat`) i zwraca ich sumę.

**Twierdzenia i przykłady** Twierdzenia definiujemy za pomocą słowa kluczowego **theorem** lub **lemma**, na przykład:

---

```
theorem add_zero (a : Nat) : add a 0 = a := -- dowód
```

---

przy czym używanie **lemma** wymaga dodania Mathliba (**import** `Mathlib.Tactic`). Jeśli mamy pewne stwierdzenie (z dowodem), którego nie będziemy używać, więc nie potrzebuje nazwy, możemy użyć słowa kluczowego **example**:

---

```
example (a : Nat) : add 0 a = a := -- dowód
```

---

Samo sformułowanie twierdzeń, lematów i przykładów jest często pewną trudnością. Możemy być więc w sytuacji, gdy chcemy najpierw sformułować tezę, a jej dowód zostawić na później. W takim przypadku nieodzowne będzie słowo kluczowe **sorry**, dzięki któremu program się skompiluje (z ostrzeżeniem), mimo braku dowodu.

---

```
theorem pow_comm (a b : Nat) : a ^ b = b ^ a :=
  sorry
```

---

Oczywiście, taka teza może być fałszywa (jak powyżej).

### Uwaga 2.1 (Plausible)

Istnieje paczka Leana o nazwie **Plausible**, która pomaga w szybkim szukaniu kontrprzykładów do hipotez. Jej działania opiera się po prostu na losowym próbkowaniu argumentów i sprawdzaniu, czy teza zachodzi. Jeśli nie, to zwraca znaleziony kontrprzykład (przez informację diagnostyczną błędu kompilacji).

---

```
import Plausible

theorem pow_comm (a b : Nat) : a ^ b = b ^ a := by
  plausible
```

---

**Aksjomaty** Czasami będziemy chcieli wprowadzić pewne aksjomaty, czyli stwierdzenia przyjmowane bez dowodu. W tym celu używamy słowa kluczowego **axiom**:

---

```
axiom add_comm (a b : Nat) : a + b = b + a
```

---

**Komentarze** Komentować kod można i należy następująco:

---

```
def foo : Nat := 42 -- to jest komentarz jednowierszowy
/- to jest
komentarz
wielowierszowy -/
```

---

**Sprawdzenie typu** Sprawdzić typ termu (czyli, przez izomorfizm Curry’ego-Howarda, również wypowiedź twierdzenia), możemy następująco:

---

```
#check Nat -- Nat : Type
#check Nat.add -- Nat.add : Nat → Nat → Nat
#check Nat.add 1 -- Nat.add 1 : Nat → Nat
#check Nat.add 1 2 -- Nat.add 1 2 : Nat
#check Nat.add_comm -- Nat.add_comm (n m : Nat) : n + m = m + n
```

---

**Problem 2.2.** Skoro `Nat` jest typu `Type`, to jaki jest typ `Type`? Jaki jest typ tego typu?

**Ewaluacja funkcji** Czasami, zamiast dowodzić twierdzeń, będziemy chcieli po prostu obliczyć wartość pewnej funkcji. Możemy to zrobić tak:

---

```
#eval Nat.add 2 3 -- 5
```

---

**Operatory potoku i odwróconego potoku** Jak na porządnym języku programowania przystało, Lean posiada operatory umożliwiające kontrolę kolejności ewaluacji wyrażeń. Operatory `<|` i `>|` zachowują się tak samo jak w F# czy Elm, sam operator `>|` znajdziemy również w Elixirze czy OCaml-u. W Haskellu odpowiadają one odpowiednio operatorom `$` i `&`. Przykłady użycia:

---

```
#eval Nat.add 1 (Nat.mul 2 3) -- 7
#eval Nat.add 1 <| Nat.mul 2 3 -- 7
#eval Nat.mul 2 3 |> Nat.add 1 -- 7
```

---

### 2.2.1. Tryby dowodzenia

Dowodząc twierdzenie, lemat, czy przykład, możemy wybrać jeden z dwóch trybów:

- *tryb termowy* (ang. *term mode*), w którym dowód jest pojedynczym (potencjalnie długim) termem,
- *tryb taktyczny* (ang. *tactic mode*), w którym dowód jest ciągiem *taktyk* modyfikujących stan dowodu.

Pierwszy jest zbliżony do rachunku lambda, drugi bardziej przypomina tradycyjne dowodzenie matematyczne. Często proste fakty łatwiej udowodnić w trybie termowym, a bardziej złożone w trybie taktycznym. Oto przykład dowodu (a właściwie zastosowania twierdzenia) w obu trybach:

---

```
-- term mode (aplikacja 2a i 2b do twierdzenia Nat.add_comm)
example (a b : Nat) : 2 * a + 2 * b = 2 * b + 2 * a := Nat.add_comm (2 * a) (2 * b)
-- tactic mode (po by następuje taktyka rw)
example (a b : Nat) : 2 * a + 2 * b = 2 * b + 2 * a := by rw [Nat.add_comm]
```

---

### 2.2.2. Argumenty i ich rodzaje

Czytelnik już zapewne zdążył zauważyć, że twierdzenia w Leanie zazwyczaj przyjmują argumenty. Na przykład w twierdzeniu `Nat.add_comm` mamy argumenty `(n : Nat)` i `(m : Nat)`, zapisywane jako `(n m : Nat)`. Takimi argumentami będą również założenia twierdzenia, na przykład w:

---

```
theorem tw (a : Nat) (b : Nat) (c : Nat) (h1 : a < b) (h2 : b < c) : a < c := sorry
```

---

Chcąc użyć takiego twierdzenia, musimy podać wszystkie argumenty, np.

---

```
axiom one_lt_two : 1 < 2
axiom two_lt_three : 2 < 3

#check tw 1 2 3 one_lt_two two_lt_three
```

---

Widać tutaj więc pewną redundancję – argumenty `a`, `b` i `c` są w pełni określone przez założenia `h1` i `h2`. Aby tego uniknąć, Lean pozwala na definiowanie argumentów *implicit*, czyli takich, które mogą być pominięte przy wywoływaniu danej funkcji lub twierdzenia. W tym celu używamy nawiasów klamrowych zamiast okrągłych:

---

```
theorem tw {a : Nat} {b : Nat} {c : Nat} (h1 : a < b) (h2 : b < c) : a < c := sorry
```

---

Teraz możemy użyć tego twierdzenia jako

---

```
#check tw one_lt_two two_lt_three
```

---

Oprócz argumentów *explicit* i *implicit*, Lean pozwala na definiowanie argumentów *instance implicit* oraz *strict implicit*, oznaczanych odpowiednio przez `[...]` oraz `{...}`. Zajmiemy się nimi później.

## 2.3. Typy Sort, Type i Prop

W ramach problemu 2.2 Czytelnik miał okazję zastanowić się nad typami **Type**, **Type 1**, **Type 2** itd. Są one częścią nieskończonej hierarchii typów **Sort**  $u$ , gdzie  $u$  jest zmienną uniwersalną oznaczającą poziom uniwersum. Typ **Sort**  $u$  jest więc typu **Sort**  $(u + 1)$ . **Type**  $u$  jest synonimem dla **Sort**  $(u + 1)$ , a **Prop** jest synonimem dla **Sort**  $0$ .

Skąd takie przesunięcie? Typy typu **Type**  $u$  (a więc sorty poziomu co najmniej 1) „trzymają” pewne obiekty matematyczne, np. term (liczba) 1 jest „trzymana” przez typ **Nat** : **Type**  $0$ . Typy typu **Prop** (czyli sorty poziomu 0) nie „trzymają” nic. Tego typu będą więc twierdzenia, od których nie wymagamy (albo nawet nie chcemy), żeby trzymały swój dowód (*proof irrelevance*).

## 2.4. Rachunek zdań i pierwsze taktyki

Od tego miejsca będziemy stosować logikę klasyczną. Jak Lean radzi sobie z rozbieżnościami między logiką intuicjonistyczną a klasyczną? Definiuje aksjomat wyboru:

```
axiom Classical.choice {α : Sort u} : Nonempty α → α,
```

z którego następnie, dzięki twierdzeniu Diaconescu, wyprowadza prawo wyłączonego środka **Classical.em** oraz wiele innych praw logiki klasycznej, jak chociażby silne prawo podwójnego przeczenia **Classical.not\_not**. Niżej omawiane taktyki Leana korzystają z tych aksjomatów, gdy jest to konieczne.

### 2.4.1. Taktyki intro, exact i apply

Jeśli cel dowodzenia znajduje się wśród założeń, to wystarczy go wskazać za pomocą taktyki **exact**. Jeśli cel jest w postaci  $P \rightarrow Q$ , to możemy użyć taktyki **intro**, aby wprowadzić do dowodu założenie  $P$  i zmienić cel na  $Q$ .

---

```
example {P : Prop} (h : P) : P := by
  exact h
```

```
example {P : Prop} : P → P := by
  intro h
  exact h
```

---

Jeśli celem jest  $Q$ , a mamy wśród założeń  $P \rightarrow Q$ , to możemy użyć taktyki **apply**, aby zmienić cel na  $P$ . Możemy stosować również aplikację znaną z typowanego rachunku lambda.

---

```
example {P Q : Prop} (h1 : P → Q) (h2 : P) : Q := by
  apply h1
  exact h2
```

```
example {P Q : Prop} (h1 : P → Q) (h2 : P) : Q := by
  exact h1 h2
```

---

### 2.4.2. Koniunkcja i alternatywa

W Lean istnieje struktura **And**, której jedynym konstruktorem jest

```
And.intro {α b : Prop} (left : α) (right : b) : α ∧ b.
```

Jako że `And.intro` ma typ  $a \rightarrow b \rightarrow a \wedge b$ , to świetnie nadaje się do użycia z `apply`, które wprowadzi dwa cele (do rozwiązania jeden po drugim).

Jeśli chcemy w ten sposób użyć konstruktora dowolnej struktury, to możemy użyć taktyki `constructor`.

---

```
example {P Q : Prop} (hp : P) (hq : Q) : P ∧ Q := by
  apply And.intro
  · exact hp
  · exact hq
```

```
example {P Q : Prop} (hp : P) (hq : Q) : P ∧ Q := by
  constructor
  · exact hp
  · exact hq
```

```
example {P Q : Prop} (h : P ∧ Q) : P := by
  exact h.Left
```

---

Podobnie istnieje struktura `Or`, której konstruktorami są

$$\text{inl} \{a b : \text{Prop}\} (h : a) : a \vee b$$

oraz

$$\text{inr} \{a b : \text{Prop}\} (h : b) : a \vee b.$$

Aby ich użyć, skorzystamy z taktyki `cases` lub `rcases` (które rekurencyjnie wywołuje `cases` zgodnie z podanym wzorem).

---

```
example {P Q R : Prop} (hpq : P ∨ Q) (hpr : P → R) (hqr : Q → R) : R := by
  cases hpq with
  | inl hp =>
    apply hpr
    exact hp
  | inr hq =>
    apply hqr
    exact hq
```

```
example {P Q R : Prop} (hpq : P ∨ Q) (hpr : P → R) (hqr : Q → R) : R := by
  rcases hpq with hp | hq
  · apply hpr
    exact hp
  · apply hqr
    exact hq
```

---

Taktyka `rcases` może się przydać również przy koniunkcjach i bardziej złożonych zdaniach, na przykład:

---

```
example {P Q R S : Prop} (h : S ∧ R ∧ P ∨ S ∧ (Q ∧ R)) : R := by
  rcases h with <_, hr, _> | <_, <_, hr>>
  · exact hr
  · exact hr
```

---

Jeśli alternatywa jest naszym celem, a nie założeniem, to przydatne będą taktyki `left` i `right`.

---

```
example {P Q : Prop} (h : P) : P ∨ Q := by
  left
  exact h
```

---

### Uwaga 2.3

Bardziej dokładnie, taktyka `constructor` używa pierwszego pasującego konstruktora struktury, a taktyki `left` i `right` używają odpowiednio pierwszego i drugiego konstruktora struktury, która ma dokładnie dwa konstruktory. Nie są one przypisane do struktur `And` i `Or`. W szczególności, zamiast `left` zawsze można używać `constructor`, ale sensowność takiego działania pozostawiamy do oceny Czytelnikowi.

### 2.4.3. Negacja

Podobnie jak w IRZ,  $\neg P$  definiujemy jako  $P \rightarrow \text{False}$ , gdzie `False` jest typem bez termów (i bez konstruktora). Bardzo łatwo jest więc dowieść, że  $\neg(P \wedge \neg P)$ .

---

```
example (P : Prop) : ¬(P ∧ ¬P) := by
  intro h
  exact h.right h.left
```

---

Z fałszu możemy oczywiście dowieść wszystko. W Leanie *ex falso quodlibet* jest realizowane przez `False.elim` lub `absurd`.

---

```
example {P Q : Prop} (h : P) (nh : ¬P) : Q := by
  exact False.elim (nh h)
```

---



---

```
example {P Q : Prop} (h : P) (nh : ¬P) : Q := by
  exact absurd h nh
```

---

### 2.4.4. Taktyki `assumption`, `contradiction`, `trivial` oraz `<;>`

Poniższy przykład (reguła *modus tollendo ponens*) Czytelnik powinien już być w stanie zrozumieć. Pokażemy, jak udowodnić go trochę prościej.

---

```
example {P Q : Prop} (h : P ∨ Q) (not_p : ¬P) : Q := by
  rcases h with hp | hq
  · exact False.elim (not_p hp)
  · exact hq
```

---

Zamiast `exact hq` możemy użyć taktyki `assumption`, która sprawdza, czy cel jest wśród założeń (nie musimy go wskazywać). Zamiast `False.elim` możemy użyć taktyki `contradiction`, która sprawdza, czy wśród założeń znajdują się dwa trywialne sprzeczne stwierdzenia.

---

```
example {P Q : Prop} (h : P ∨ Q) (not_p : ¬P) : Q := by
  rcases h with hp | hq
  · contradiction
  · assumption
```

---



Jeśli nie używamy wprowadzonych założeń (w przykładzie `hp` i `hq`), to nie musimy ich nazywać – dalej będą one dostępne dla taktyk `assumption` i `contradiction`.

---

```
example {P Q : Prop} (h : P ∨ Q) (not_p : ¬P) : Q := by
  cases h
  · contradiction
  · assumption
```

---

Istnieje również taktyka `trivial`, która próbuje rozwiązać cel używając kilku prostych taktyk, jak `assumption`, `contradiction` czy `rfl` (użyteczna przy równościach).

---

```
example {P Q : Prop} (h : P ∨ Q) (not_p : ¬P) : Q := by
  cases h
  · trivial
  · trivial
```

---

Jeśli chcemy użyć tej samej taktyki dla różnych przypadków, to możemy użyć `<;>` w następujący sposób:

---

```
example {P Q : Prop} (h : P ∨ Q) (not_p : ¬P) : Q := by
  cases h <;> trivial
```

---

**Problem 2.4.** Zrobić pierwszy zestaw zadań:

---

```
-- Zadanie 1
example {A B C D : Prop} (h1 : A → B) (h2 : B → C) (h3 : C → D) : A → D := sorry

-- Zadanie 2
example (P Q R : Prop) (h1 : P ∧ Q) (h2 : P → R) : R := sorry

-- Zadanie 3
theorem modus_tollens {P Q : Prop} (h1 : P → Q) (h2 : ¬Q) : ¬P := sorry

-- Zadanie 4
example {P Q : Prop} (h : ¬(P ∨ Q)) : ¬P ∧ ¬Q := sorry

-- Zadanie 5
example {P Q : Prop} (h : ¬P ∧ ¬Q) : ¬(P ∨ Q) := sorry

-- Zadanie 6
example (A B C D E : Prop) (h1 : A ∧ B) (h2 : B → ¬C ∧ D) (h3 : E → C) : ¬E := sorry

-- Zadanie 7
example {P : Prop} : ((P → P) → P) → P := sorry

-- Zadanie 8
-- Pokaż, że system logiczny oparty na logice klasycznej z aksjomatem A jest trywialny.
axiom A {P : Prop} : (((P → P) → P) → P) → P

example {Q : Prop} : Q := sorry
```

---

### 3. Literatura

- [1] M. H. Sørensen, P. Urzyczyn, Lectures on the Curry-Howard Isomorphism, *Studies in Logic and the Foundations of Mathematics*, vol. 149, Elsevier, 2006.